

CS321 F'04 Lecture Notes
Lecture 6

Syntax Analysis

Specify legal program formats using **context-free grammar (CFG)**


- Use **Backus-Naur Form (BNF)** as notation.
- Gives precise, readable specification.
- Many CFG's have efficient **parsers**.
- Parser **recognizes** syntactically legal programs and **rejects** illegal ones.

Successful parse also captures **hierarchical** structure of programs (expressions, blocks, etc.).

- Convenient representation for further semantic checking (e.g., types) and for code generation.

We can use another program generator (e.g., *yacc* or *CUP*) to generate parser automatically from grammar.

Syntax Analysis (Parsing)



Grammars

A **Context-free Grammar** is described by a set of **productions** (also called **rewrite rules**):

Examples:

$stmt \rightarrow if\ expr\ then\ stmt\ else\ stmt$

$expr \rightarrow expr + expr \mid expr * expr \mid (expr) \mid -expr \mid id$

Grammars contain **terminals** (\equiv **tokens**) (e.g. *id*) and **non-terminals** (e.g., *expr*).

Grammars have a distinguished **start symbol** (ordinarily listed first, e.g., *stmt*).


The language **generated** by a grammar is the set of **sentences** (strings) of terminals that can be **derived** by repeated application of productions, beginning with **start symbol**.

- We write $L(G)$ for the language generated by grammar G .

Parse Trees

Graphical representation of a derivation.

Example tree for derivation of sentence $-(id + id)$:



Each application of a production corresponds to an **internal** node, labeled with a **non-terminal**.

Leaves are labeled with **terminals**.

The derived sentence is found by reading leaves left-to-right.

Derivations

Can “linearize” a parse tree into a sequence of one-step derivations.

Example:

$expr \Rightarrow -\ expr$
 $\Rightarrow -(expr)$
 $\Rightarrow -(expr + expr)$
 $\Rightarrow -(id + expr)$
 $\Rightarrow -(id + id)$

or
 $expr \overset{*}{\Rightarrow} -(id + id)$

Here \Rightarrow is read “derives” and $\overset{*}{\Rightarrow}$ is read “derives in zero or more steps.”

This example gives a **leftmost derivation**, i.e., at each step, the leftmost non-terminal is replaced.

Can define **rightmost derivation** analogously:

$expr \Rightarrow \dots$
 $\Rightarrow -(expr + expr)$
 $\Rightarrow -(expr + id)$
 $\Rightarrow -(id + id)$

Ambiguity


Every parse tree has a **unique** leftmost derivation and a **unique rightmost** derivation.

BUT a given **sentence** in $L(G)$ can have more than one parse tree. Grammars G for which this is true are called **ambiguous**.

Example: with our grammar, the sentence

$id + id * id$

has two parse trees:



We may think of the left tree as being the “correct” one, but nothing in the grammar says this.

To avoid the problems of ambiguity, we can:

- Rewrite grammar
- Use “disambiguating rules” when we implement parser for grammar.

A classic ambiguity: the “dangling else”


Suppose we want **else** clauses to be optional in **if** statements. Here’s a possible grammar:

$stmt \rightarrow \text{if } expr \text{ then } stmt$
 $\quad | \text{if } expr \text{ then } stmt \text{ else } stmt \mid \dots$

But with this grammar the sentence

$\text{if } E_1 \text{ then if } E_2 \text{ then } S_1 \text{ else } S_2$

has two possible parse trees:



Resolving Ambiguity by Rewriting Grammar


Usually want the first tree (else goes with most recent then), but grammar is ambiguous.

Solution: rewrite grammar using new non-terminals *mst* (“matched statement”) and *ust* (“unmatched statement”).

```

stmt → mst | ust
mst → if expr then mst else mst
      | ...
ust → if expr then stmt
      | if expr then mst else ust
    
```

Now only one parse is possible. Assuming *S*₁, *S*₂ are not unmatched if statements:




Rewriting Arithmetic Grammars

Can build precedence/associativity into grammar using extra non-terminals, e.g.

```

atom → (expr) | id
primary → -primary | atom
factor → primary ↑ factor | primary
term → term * factor | term / factor | factor
expr → expr + term | expr - term | term
    
```

Example: *id* * *id* + *id* ↑ *id* ↑ *id*



Ambiguity in Arithmetic Expressions

A grammar such as

```

E → E + E | E - E | E * E | E / E
     E ↑ E | (E) | - E | id
    
```

is ambiguous about order of operations.

Want to define

- **Precedence** - which operation is done first?

- **Associativity** - is

*X op*₁ *Y op*₂ *Z*

equivalent to (*X op*₁ *Y*) *op*₂ *Z* or to *X op*₁ (*Y op*₂ *Z*) (assuming *op*₁ and *op*₂ have same precedence).

The “usual” rules (based on common usage in written math) give the following precedences, highest first:

- (unary minus)
- ↑ (exponentiation)
- * /
- + -

All the binary operators are left-associative except exponentiation (↑).

We can handle precedence/associativity information as “side-conditions” to ambiguous grammar when building a parser (by hand or via a parser generator).

Extended BNF

Various semi-standard extensions to BNF are often used in language manuals.

They allow grammar specifications to avoid explicit recursion and ϵ -productions, by adding **optional** symbols, **repetition**, and **grouping**.

```

[a] means a |  $\epsilon$ 
{a} means  $\epsilon$  | a | aa | ...
(a | b)c means ac | bc
    
```

To avoid confusion between new meta-symbols and terminals, we often enclose the latter in single quotes:

Example:

```

atom → '( ' expr ' )' | id
primary → -primary | atom
factor → primary { ↑ primary }
term → { factor ( * | / ) } factor
expr → { term ( + | - ) } term
    
```

Must be careful not to **change** the generated language accidentally when going between EBNF and BNF.

Also, note that EBNF grammar given in example is ambiguous because we’ve lost the associativity information present in the BNF version (even though both generate same language).

Properties of CFG's

Any **regular** language can be described by a CFG.

Example: $(a | b)^* abb$

$$A_0 \rightarrow aA_0 | bA_0 | aA_1$$
$$A_1 \rightarrow bA_2$$
$$A_2 \rightarrow bA_3$$
$$A_3 \rightarrow \epsilon$$

But we don't use CFG's for lexical analysis, because it's overkill.

Regular expressions are:

- easier to understand
- shorter
- always lead to efficient analyzers

Any CFL can be parsed by a computer program, but only **some** CFL's can be parsed **efficiently**.

We'll study both "bottom-up" and "top-down" parsing methods.

CFG's can't do everything

Not every language is a CFL.

Example: $L = \{wcw \mid w \in (a | b)^*\}$ is not CF.

- L abstracts idea of variable declaration before use.
- So "semantic" analysis (type-checking) uses additional (mostly ad-hoc) techniques.